
                                   
INTRODUCTION

A general purpose resonator routine is implemented
using a PIC17C42. This routine is used to generate
multiple tones. A tone signal is normally generated
using extensive table lookup schemes. When a multiple
tone signal is desired, each tone must have its own
lookup table, thus requiring a large amount of storage
space, especially when various frequencies are to be
generated. This application note implements tone gen-
eration using recursive techniques. The algorithm for a
resonator is developed and implemented using
PIC17C42.

THEORY 

Generation of a single tone basically implies generating
samples of a sine/cosine wave. The Z-Transform of a
sine wave is given as follows:

The impulse response of the above transform (i.e., for
X(z) = 1) will generate a sine wave of frequency w
sampled at a rate of T (= 1/fs). Thus the above equation
is translated to:

The above equation can be rewritten in a difference
equation form as follows:

y(n) - 2y(n-1)cos(wT) + y(n-2) = x(n-1)sin(wT)

Rearranging the above equation and setting, x(n) as an
impulse sequence, the following recursive equations
are obtained.

y(n) = 2*K1*y(n-1)-y(n-2)
y(n-2) = y(n-1)
y(n-1) = y(n)

with the following conditions:

K1 = cos(wT)
K2 = initial y(n-1) = sin(wT)
K3 = initial y(n-2) = 0

IMPLEMENTATION

The above developed algorithm is implemented as a
subroutine using a PIC17C42. All computations are
performed using double precision arithmetic
(16/32-bits). The recursive tone generation algorithm is
implemented as a subroutine (labeled
as Resonator ). This subroutine generates samples
of a single tone. To generate multiple frequencies, sim-
ply call this resonator routine for the desired frequen-
cies and sum the outputs. The three tone co-efficients
are stored in program memory and are read into the
data memory using TABLRD instructions. 

The fully commented code is listed in Appendix A. The
timing and memory requirements are included in the
comment sections of the code. For a listing of the
header file 17C42.h  and the macro definition file
17C42.mac  please refer to Appendices C and D
respectively of the application note DS00540. This
code can be easily modified and used in various
applications like DTMF generation, sound generation,
etc. The tones generated can easily be output to an
on-chip PWM channel which in turn can drive a speaker
for producing various sounds. If using a PWM channel,
set the PWM frequency much higher than the sampling
frequency used (in the example code, for 8 kHz sam-
pling frequency, use at least 20 kHz PWM frequency).

As an example, a dual tone is generated and the result-
ing digital wave form is analyzed. The main program
calls the function Resonator  twice for generating the
two desired tones and the two outputs are summed. A
sampling frequency of 8 kHz was used to generate a
dual tone of 800 Hz and 1.10 kHz. The resulting wave
form is shown in Figure 1. The spectrum of the signal
shown in Figure 2 shows two peaks corresponding to
the two desired tones (800 Hz and 1.10 kHz). The
assembly code was tested using PICMASTER® (Micro-
chip’s Universal In-Circuit Emulator System). 
Tone Generation

The generated tones were captured into the PICMASTER’s trace buffer and then transferred to Microsoft  Excel  using
Dynamic Data Exchange (DDE). Once the data is in Excel it is analyzed using Excel's FFT utility.

FIGURE 1: DUAL TONE WAVE FORM CAPTURED BY PICMASTER

FIGURE 2: FREQUENCY SPECTRUM SHOWING THE DUAL TONE FREQUENCIES

PERFORMANCE

Table 1 below provides the performance of the resona-
tor (labeled in the source code as Resonator ) in 
terms of both timing and memory requirements. Since 
a double precision multiplier is used (software imple-
mentation), the multiplier timing is not always constant. 
Therefore the timings are given for the worst case. 
Note that irrespective of the frequency and the sam-
pling (resolution) of the tone, program memory require-
ments is only 54 locations which in case of table 
lookup could be very large.

TABLE 0-1:

Cycles 235 Cycles
Time @ 16 MHz 58.75 µs
Time @ 25 MHz 47 µs
Data Memory 9 + 9* (# of tones to be generated)
Program memory 54 locations
DS00543C-page 2
APPLICATIONS

Tone generation is required in many applications. The
code provided in this application note may be used as
a general purpose routine to generate desired tones. It
can be used in applications involving secure off-site
control, where commands/data in the format of tones
are transmitted over a telephone line. The tone genera-
tion finds applications involving signal modulations as
well. The routine can be used to generate audible tones
and output to a speaker connected to an I/O Port or a
PWM channel.
  1997 Microchip Technology Inc.

APPENDIX A: TONE.ASM 
MPASM 01.40 Released             TONE.ASM   1-16-1997  15:04:23         PAGE  1

LOC  OBJECT CODE     LINE SOURCE TEXT
  VALUE

                    00001         LIST    P=17C42, columns=120,WRAP, R=DEC
                    00002 ;
                    00003 ;***************************************************************
                    00004 ;                       Dual Tone Generation
                    00005 ;
                    00006 ;       A generic resonator subroutine is implemented to generate
                    00007 ; tones. Samples Of A Sin/Cos Wave are generated using
                    00008 ; recursive techniques. Table Lookups are thus avoided
                    00009 ; This is especially useful in generating multiple tones
                    00010 ; (e.g. DTMF tone generation, tone signalling, etc), or programmable 
                    00011 ; tone generation which may vary for each application unit.
                    00012 ;
                    00013 ;
                    00014 ;       Program:          TONE.ASM 
                    00015 ;       Revision Date:   
                    00016 ;                         1-13-97      Compatibility with MPASMWIN 1.40
                    00017 ;
                    00018 ;*************************************************************
                    00019 ;
                    00020         include “p17c42.inc”
                    00001         LIST
                    00002 ;P17C42.INC Standard Header File, Ver. 1.03 Microchip Technology, Inc.
                    00264         LIST
                    00021 
                    00022         #define TRUE    1
                    00023         #define FALSE   0
                    00024         #define LSB     0
                    00025         #define MSB     7
                    00026 
                    00027         include “17c42.mac”
                    00945         LIST
                    00028 ;
                    00029 ;*******************************************************************
                    00030 ;       TBLADDR
                    00031 ;
                    00032 ; DESCRIPTION:
                    00033 ;       Load 16 bit table pointer with specified label
                    00034 ;
                    00035 ; TIMING (in cycles):
                    00036 ;       4
                    00037 ;
                    00038 
                    00039 TBLADDR MACRO   label
                    00040 
                    00041         MOVLW   (label)
                    00042         MOVWF   TBLPTRL
                    00043         MOVLW   page    (label)
                    00044         MOVWF   TBLPTRH
                    00045 
                    00046         ENDM
                    00047 
                    00048 ;********************************************************************
                    00049 ;       ADDLBL
                    00050 ;

Please check the Microchip BBS for the latest version of the source code. 
Microchip’s Worldwide Web Address: www.microchip.com; Bulletin Board Support: MCHIPBBS using CompuServe® 
(CompuServe membership not required).

                    00051 ; DESCRIPTION:
                    00052 ;       Add A Label (16 bit constant) To A File Register (16 bit)
                    00053 ;
                    00054 ; TIMING (in cycles):
                    00055 ;       4
                    00056 ;
                    00057 
                    00058 ADDLBL  MACRO   label,f
                    00059 
                    00060         MOVLW   (label)
                    00061         ADDWF   f+BB0, F
                    00062         MOVLW   page    (label)
                    00063         ADDWFC  f+BB1, F
                    00064 
                    00065         ENDM
                    00066 
                    00067 ;*****************************************************************
                    00068 
                    00069         LIST
                    00070 ;
                    00071         CBLOCK  0
  00000000          00072                 BB0,BB1,BB2,BB3 ; RAM offset constants
                    00073         ENDC
                    00074 ;
                    00075         CBLOCK  0x18
  00000018          00076          AARG,AARG1                    ; 16 bit multiplier A
  0000001A          00077          BARG,BARG1                    ; 16 bit multiplicand B
  0000001C          00078          DPX,DPX1,DPX2,DPX3            ; 32 bit multiplier result = A*B
                    00079          ENDC
                    00080 ;
                    00081         CBLOCK                         ; Generic Resonator Variables
  00000020          00082          K1,K11                        ; K1
  00000022          00083          SinCos_2C,SinCos_2,SinCos1_2  ; y(n-2), Init Value = K3
  00000025          00084          SinCos_1,SinCos1_1            ; y(n-1), Init Value = K2
  00000027          00085          SinCos,SinCos1                ; y(n)
                    00086                 
                    00087         ENDC
                    00088 
                    00089         CBLOCK
  00000029          00090          tone1K1,tone1K11              ; Tone 1 variables
  0000002B          00091          tone1_2C,tone1_2,tone11_2
  0000002E          00092          tone1_1,tone11_1
  00000030          00093                 tone1,tone11
                    00094 
  00000032          00095          tone2K1,tone2K11             ; Tone 2 variables
  00000034          00096          tone2_2C,tone2_2,tone21_2
  00000037          00097          tone2_1,tone21_1
  00000039          00098          tone2,tone21
                    00099 
  0000003B          00100           dualTone,dualTone1          ; tone1+tone2
                    00101         ENDC
                    00102 
                    00103 ;
                    00104 ;************************************************************
                    00105 ;
                    00106 #define ADD_OFFSET  FALSE
                    00107 ;
                    00108 #define coeff_addr  0x0030  ; prog mem addr of sine wave frequency,etc
                    00109 #define DummyAddr   0x0A00
                    00110 ;
                    00111 ;******************************************************************
                    00112 ;               Test Routine For Sin Wave Generation
                    00113 ;
                    00114 ;       (a) Call Initialization Subroutine to read desired sine wave
                    00115 ;               freq from prog mem to data mem.
                    00116 ; (b) Call the resonator subroutine to generate samples of desired
DS00543C-page 4   1997 Microchip Technology Inc.

                    00117 ; sine wave.
                    00118 ; (c) Perform a Dummy Table Write of The Sine Wave Data, so that
                    00119 ; PIC-MASTER emulator can capture the data into trace buffer
                    00120 ;       
                    00121 ;******************************************************************
                    00122 ;
0000                00123         ORG     0x0000
0000 C040           00124         goto    start
                    00125 ;
                    00126 ;******************************************************************
                    00127 ;               Resonator Co-efficients For Desired Tones
                    00128 
0030                00129    ORG     coeff_addr
                    00130         
                    00131 ; Tone 1 Resonator Constants
                    00132 ; Sample Rate = fs = 8 khz, Tone Freq = f = 0.800 Khz
                    00133 ;
0030 678E           00134   DATA    26510  ; K1 = COS(wT) = COS(360*f/fs)
0031 259E           00135   DATA    9630   ; K2 = SIN(wT) = SIN(360*f/fs) = init value of y(n-1)
0032 0000           00136   DATA    0      : K3 is init value of y(n-2)
                    00137 ;
                    00138 ; Tone 2 Resonator Constants
                    00139 ; Sample Rate = fs = 8 khz, Tone Freq = f = 1.10 Khz
                    00140 ;
0033 5321           00141   DATA    21281  ; K1 = COS(wT) = COS(360*f/fs)
0034 1855           00142   DATA   6229    ; K2 = SIN(wT) = SIN(360*f/fs) = init value of y(n-1)
0035 0000           00143   DATA    0      ; K3 is init value of y(n-2)
                    00144 ;
                    00145 ;******************************************************************
                    00146 ;
0040                00147   ORG     0x0040
0040                00148 start
0040 E057           00149    call    Coeff_Read
                    00150 ; load table pointers with a dummy  addr for PIC-MASTER data capture
                    00151         MOVK16  DummyAddr,TBLPTRL
                        M 
0041 B000               M   MOVLW   (0x0A00) & 0xff
0042 010D               M   MOVWF   TBLPTRL+B0
0043 B00A               M   MOVLW   ((0x0A00) >> 8)
0044 010E               M   MOVWF   TBLPTRL+B1
                        M 
0045                00152 NextSample
0045 0000           00153   nop
0046 A43B           00154   tlwt    0,dualTone
0047                00155 capture
0047 AE3C           00156   tablwt  1,0,dualTone+BB1   ; for PIC-MASTER trace capture
0048 0000           00157   nop
                    00158 ;
0049 B029           00159   movlw   tone1K1            ; load indirect address pointer
004A 0101           00160   movwf   FSR0               ; for Tone 1
004B E06C           00161   call    Resonator          ; Compute next sample of tone 1
                    00162 
004C B032           00163   movlw   tone2K1            ; load indirect address pointer
004D 0101           00164   movwf   FSR0               ; for Tone 2
004E E06C           00165   call    Resonator          ; compute next sample of tone 2
                    00166 ;
                    00167 ; Compute Tone1 + Tone2 for dual tone generation
                    00168 ;
                    00169   ADD16ACC tone1,tone2,dualTone  ; dualTone = tone1 + tone2
                        M 
004F 6A30               M   movfp   tone1+B0,WREG
0050 0E39               M   addwf   tone2+B0,W
0051 013B               M   movwf   dualTone+B0
0052 6A31               M   movfp   tone1+B1,WREG
0053 103A               M   addwfc  tone2+B1,W
0054 013C               M   movwf   dualTone+B1
  1997 Microchip Technology Inc. DS00543C-page 5

                        M 
                    00170 ;
0055 C045           00171    goto    NextSample
                    00172 ;
0056 C056           00173 self  goto self
                    00174 ;
                    00175 ;*****************************************************************
                    00176 ; Initialization routine :
                    00177 ; Read Tone 1 & Tone 2 Resonator Frequencies from Program Memory To
                    00178 ; Data Memory 
                    00179 ;
                    00180 ;       Program Memory  :       3 + 8*(#of tones to be generated)
                    00181 ;       Timing          :       4 + 11*(#of tones to be generated)
                    00182 ;**************************************************************
                    00183 
0057                00184 Coeff_Read
                    00185 
                    00186    TBLADDR coeff_addr
                        M 
0057 B030               M    MOVLW   (0x0030)
0058 010D               M    MOVWF   TBLPTRL
0059 B000               M    MOVLW   page    (0x0030)
005A 010E               M    MOVWF   TBLPTRH
                        M 
005B A929           00187    tablrd  0,1,tone1K1
005C A029           00188    tlrd    0,tone1K1
005D AB2A           00189    tablrd  1,1,tone1K1+BB1 ; read K1
005E A02E           00190    tlrd    0,tone1_1
005F AB2F           00191    tablrd  1,1,tone1_1+BB1 ; read K2
0060 A02C           00192    tlrd    0,tone1_2
0061 A22D           00193    tlrd    1,tone1_2+BB1   ; read K3
0062 292B           00194    clrf    tone1_2C, F
                    00195 
0063 A932           00196    tablrd  0,1,tone2K1
0064 A032           00197    tlrd    0,tone2K1
0065 AB33           00198    tablrd  1,1,tone2K1+BB1 ; read K1
0066 A037           00199    tlrd    0,tone2_1
0067 AB38           00200    tablrd  1,1,tone2_1+BB1 ; read K2
0068 A035           00201    tlrd    0,tone2_2
0069 A236           00202    tlrd    1,tone2_2+BB1   ; read K3
006A 2934           00203    clrf    tone2_2C, F
                    00204 
006B 0002           00205    return
                    00206 ;
                    00207 ;*********************************************************
                    00208 ;                       Resonator Subroutine
                    00209 ;
                    00210 ; Before calling this routine, load the indirect register, FSR0
                    00211 ; with the starting RAM address of the desired Tone Variables, 
                    00212 ; ( eg. For Tone1 Generation, load FSR0 with “Tone1K1” address.
                    00213 ;
                    00214 ; 
                    00215 ;
                    00216 ; Timing (worst case) :
                    00217 ;               20 + 36 + (worst case multiplier time)
                    00218 ;               = 56 + 179 = 235 cycles 
                    00219 ;               = 58.75 uS @ 16Mhz
                    00220 ;               = 47.00 uS @ 20Mhz
                    00221 ;               = 37.60 uS @ 25 Mhz
                    00222 ;
                    00223 ; Memory Requirements :
                    00224 ;               Program Memory  : 54 locations
                    00225 ;               Data Memory     : 9 + 9*(#of tones to be generated)
                    00226 ;
                    00227 ;************************************************************
                    00228 ;
DS00543C-page 6   1997 Microchip Technology Inc.

                    00229 
006C                00230 Resonator
                    00231         
                    00232 ;
                    00233 ; Transfer tone variables to resonator’s variables using indirect 
                    00234 ; addressing This is necessary for making the “Resonator” a generic
                    00235 ; subroutine, so that the same code can be called for various
                    00236 ; tones.
                    00237 ; Indirect addressing mode can be used throught the code in this 
                    00238 ; subroutine, but is less efficient.
                    00239 ;
006C 8D04           00240   bcf     ALUSTA,FS1
006D 8404           00241   bsf     ALUSTA,FS0              ; auto increment FSR0
006E 4020           00242   movpf   INDF0,K1+BB0
006F 4021           00243   movpf   INDF0,K1+BB1
0070 4022           00244   movpf   INDF0,SinCos_2C
0071 4023           00245   movpf   INDF0,SinCos_2+BB0
0072 4024           00246   movpf   INDF0,SinCos_2+BB1
0073 4025           00247   movpf   INDF0,SinCos_1+BB0
0074 4026           00248   movpf   INDF0,SinCos_1+BB1
0075 4027           00249   movpf   INDF0,SinCos+BB0
0076 4028           00250   movpf   INDF0,SinCos+BB1
                    00251 
                    00252 ;
                    00253 ; Compute 2*K1*y(n-1)
                    00254 ;
                    00255   MOVFP16 K1,BARG         
                        M 
0077 7A20               M   MOVFP   K1+B0,BARG+B0           ; move A(B0) to B(B0)
0078 7B21               M   MOVFP   K1+B1,BARG+B1           ; move A(B1) to B(B1)
                        M 
                    00256   MOVFP16 SinCos_1,AARG
                        M 
0079 7825               M   MOVFP   SinCos_1+B0,AARG+B0     ; move A(B0) to B(B0)
007A 7926               M   MOVFP   SinCos_1+B1,AARG+B1     ; move A(B1) to B(B1)
                        M 
007B E0A2           00257   call    DblMult                 
                    00258   RLC32   DPX                     ; DPX = 2*K1*y(n-1)
                        M 
007C 8804               M   BCF     ALUSTA,C
007D 1B1C               M   RLCF    DPX+B0, F
007E 1B1D               M   RLCF    DPX+B1, F
007F 1B1E               M   RLCF    DPX+B2, F
0080 1B1F               M   RLCF    DPX+B3, F
                        M 
                    00259 ;
                    00260 ; subtract y(n-2)*(2**15)
                    00261 ;
0081 2922           00262   clrf    SinCos_2C, F
                    00263   RRC24   SinCos_2C
                        M 
0082 1A24               M   RLCF    SinCos_2C+B2,W          ; move sign into carry bit
0083 1924               M   RRCF    SinCos_2C+B2, F
0084 1923               M   RRCF    SinCos_2C+B1, F
0085 1922               M   RRCF    SinCos_2C+B0, F
                        M 
                    00264   SUB24   SinCos_2C,DPX1          ; DPX = 2*K1*y(n-1) - y(n-2)
                        M 
0086 6A22               M   MOVFP   SinCos_2C+B0,WREG       ; get lowest byte of a into w
0087 051D               M   SUBWF   DPX1+B0, F              ; sub lowest byte of b, save in b(B0)
0088 6A23               M   MOVFP   SinCos_2C+B1,WREG       ; get 2nd byte of a into w
0089 031E               M   SUBWFB  DPX1+B1, F              ; sub 2nd byte of b, save in b(B1)
008A 6A24               M   MOVFP   SinCos_2C+B2,WREG       ; get 3rd byte of a into w
008B 031F               M   SUBWFB  DPX1+B2, F              ; sub 3rd byte of b, save in b(B2)
                        M 
                    00265 

                    00266   RLC24   DPX1                    ; adjust decimal point
                        M 
008C 8804               M   BCF     ALUSTA,C
008D 1B1D               M   RLCF    DPX1+B0, F
008E 1B1E               M   RLCF    DPX1+B1, F
008F 1B1F               M   RLCF    DPX1+B2, F
                        M 
                    00267 ;
                    00268 ; update past samples with newly computed values
                    00269 ;
                    00270   MOVPF16 DPX2,SinCos             ; y(n) = 2*K1*y(n-1) - y(n-2)
                        M 
0090 5E27               M   MOVPF   DPX2+B0,SinCos+B0       ; move A(B0) to B(B0)
0091 5F28               M   MOVPF   DPX2+B1,SinCos+B1       ; move A(B1) to B(B1)
                        M 
                    00271   MOV16   SinCos_1,SinCos_2       ; y(n-2) = y(n-1)
                        M 
0092 6A25               M   MOVFP   SinCos_1+B0,WREG        ; get byte of a into w
0093 0123               M   MOVWF   SinCos_2+B0             ; move to b(B0)
0094 6A26               M   MOVFP   SinCos_1+B1,WREG        ; get byte of a into w
0095 0124               M   MOVWF   SinCos_2+B1             ; move to b(B1)
                        M 
                    00272   MOVPF16 DPX2,SinCos_1           ; y(n-1) = y(n)
                        M 
0096 5E25               M   MOVPF   DPX2+B0,SinCos_1+B0     ; move A(B0) to B(B0)
0097 5F26               M   MOVPF   DPX2+B1,SinCos_1+B1     ; move A(B1) to B(B1)
                        M 
                    00273 ;
                    00274 ; Generation Of The Next Sample Of The Resonator (sine wave) is done.
                    00275 ; The 16 bit result is stored in location “SinCos” (low Byte) &
                    00276 ; “SinCos+1” (High Byte)
                    00277 ;
                    00278 ; write back all the computed values to respective tone variables
                    00279 ;
0098 0701           00280   decf    FSR0, F
0099 8D04           00281   bcf     ALUSTA,FS1
009A 8C04           00282   bcf     ALUSTA,FS0
009B 6028           00283   movfp   SinCos+BB1,INDF0
009C 6027           00284   movfp   SinCos+BB0,INDF0
009D 6026           00285   movfp   SinCos_1+BB1,INDF0
009E 6025           00286   movfp   SinCos_1+BB0,INDF0
009F 6024           00287   movfp   SinCos_2C+BB2,INDF0
00A0 6023           00288   movfp   SinCos_2C+BB1,INDF0
                    00289 ;
00A1 0002           00290   return
                    00291 ;
                    00292 ;*****************************************************************
                    00293 ;       Include Double Precision Multiplication Routine
                    00294 ;*****************************************************************
  00000001          00295 SIGNED  equ     TRUE
                    00296 
                    00297    include “17c42mpy.mac”
                    00178    LIST
                    00298 
                    00299 ;
                    00300    END
DS00543C-page 8   1997 Microchip Technology Inc.

MEMORY USAGE MAP (‘X’ = Used,  ‘-’ = Unused)

0000 : X--------------- ---------------- ---------------- XXXXXX----------
0040 : XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX
0080 : XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX
00C0 : XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX
0100 : XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX
0140 : XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX
0180 : XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX XXXX------------

All other memory blocks unused.

Program Memory Words Used:   379

Errors   :     0
Warnings :     0 reported,     0 suppressed
Messages :     0 reported,     0 suppressed
  1997 Microchip Technology Inc. DS00543C-page 9

• Neither Microchip nor any other semiconductor manufacturer can guarantee the security of their code. Code protection does not 

mean that we are guaranteeing the product as “unbreakable”.
• Code protection is constantly evolving. We at Microchip are committed to continuously improving the code protection features of 

our product.

If you have any further questions about this matter, please contact the local sales office nearest to you.

